Complexity Analysis of Climate Data during Last Deglaciation

Maria Yu Qian Lim
500386873

Abstract

This report consists of seven sections. In the first section, a brief introduction of events in last deglaciation period, importance of studying paleoclimate records and some previous work findings are discussed. The second section describes the dataset used in this project, and followed by third section where preprocessing of data is showed. The main results of this project are found in Section 4 and 5, where complexity of time series is analysed and critical slowing down for abrupt climate changes is indicated respectively. In Section 6, a comparative analysis with linear data is performed. The report is wrapped up by a conclusion and suggestion for future work in the last section.

1. Introduction

There were many events happened during the last deglaciation period [1], such as the Dansgaard Oeschger or DO oscillations, Bolling Allerod warming, and Younger Dryas. D-O oscillations are the most dramatic and frequent abrupt climate changes in the geological record, and studies had proven that it is not just a regional phenomena but hemispheric-wide effect.[2] Bolling Allerod interstadial ran from 14,690 to 12,890 years before the present (BP). [3] It began with the end of the cold period known as the Oldest Dryas, and ended abruptly with the onset of the Younger Dryas. On the other hand, Younger Dryas which happened from around 12,900 to 11,700 years BP , was a return to glacials condition which temporarily reversed the climatic warming.[4] This temperature change occurred at the end of the Pleistocene epoch and immediately before the current, warmer Holocene epoch. In this study the latter two events are focused, same as many other previous studies on paleoclimate records.

The purpose of studying the switching between one climate state and another is to, at least, try to understand the underlying dynamics of the global climate system, although they are still unclear. [5] It is concerned that human activities that speed up the global warming could tip the climate out of its current stable regime since the beginning of Holocene interglacial epoch. Many studies had been carried out on the paleoclimate records to detect early warning signals before a change of system state happened. Most studies focused on two abrupt warming events during the last deglaciation, and the interested period of study is usually from the Last Glacial Maximum (LGM) [6] to the end of the Younger Dryas or to the start of the Bølling-Allerød. For example, Lenton and Livina found that both autocorrelation function (ACF) and detrended fluctuation analysis (DFA) indicators and variance show an increasing trend to the end of the Younger Dryas. [c] Using autocorrelation at lag 1 method, Dakos et al. observed a marked increase in slowing down before the end of Younger Dryas but only weak signs of slowing down for Bolling Allerod warming. [b]

2. Dataset

The Greenland Ice Sheet Project 2 (GISP2) ice core temperature data from National Geophysical Data Center is used. [7] The raw climate record consists of 1632 data points, from 50 000 years before present until present. Note that before present is a time scale used mainly in disciplines such as archaeology, where “present” is usually referred to 1 Jan 1950.

In[ ]:=
```
data = Reverse[Import[
    "C:\\Users\\maria\\OneDrive\\Documents\\1USyd\\2nd_sem\\CSYS5040\\
      Assessment\\Assessment 3\\greenland.xlsx",
    {"Dataset", 1}, "HeaderLines" → 1]]
```

Out[ ]=

Age	Temperature (C)
49.981	−39.9694
49.922	−39.8165
49.868	−40.1018
49.806	−40.2651
49.746	−40.5505
49.69	−40.4792
49.635	−40.6523
49.579	−41.0297
49.521	−41.152
49.47	−41.4679
49.418	−41.5596
49.358	−41.3661
49.293	−41.4373
49.228	−41.6921
49.167	−43.1498
49.095	−43.7514
49.026	−43.5881
48.964	−42.8847
48.901	−42.8135
48.833	−43.1905

rows 1–20 of 1632

In[ ]:=
```
AgeRaw = data[All, "Age"];
TemperatureRaw = data[All, "Temperature (C)"];
Length[AgeRaw]
Length[TemperatureRaw]
tsRaw = TimeSeries[
   Table[{AgeRaw[[i]], TemperatureRaw[[i]]}, {i, Length[TemperatureRaw]}]]
Show[{
   ListLinePlot[tsRaw, ScalingFunctions → {"Reverse", Identity}, PlotRange → All,
     AxesLabel → {RawBoxes["time, x10^3 yrs BP"], RawBoxes["Temperature,C"]},
     PlotLabel → "Figure 1 GISP2 ice core (raw data time series)",
     LabelStyle → {GrayLevel[0], Bold}, ImageSize → 800],
   Plot[-60, {x, -12.9, -11.7}, PlotStyle → Blue, Filling → Axis],
   Plot[-60, {x, -14.69, -12.89}, PlotStyle → Red, Filling → Axis]
  }]
```

Out[ ]= 1632

Out[ ]= 1632

Out[ ]= TimeSeries[Time: 0.0951 to 50. Data points: 1586]

Out[ ]=



The red band marks the Bølling-Allerød warming and the blue the Younger Dryas. The period for these two climate events are based on the time given on Wikipedia. It is important to check if the climate data is regularly-sampled before processing or carrying out any further analysis. In Mathematica, it is just one step to confirm with this, which is shown below. The result shows that the data in fact is not equidistant.

In[ ]:=
```
RegularlySampledQ[tsRaw]
```

Out[ ]=
```
False
```

3. Pre-processing and preparation of data

The climate data is resampled and interpolated so that it is transformed to a temporally consistent time series, and ensured by the function “RegularlySampledQ[]”. The processed data consists of 1996 data points starting from 0.0951 to 50 thousand ages before present, with a time step of 0.025 thousand ages. By this, the record is then presented in its data points itself to be better handled in various analyses which are discussed in following sections. Another significant step in time series processing is to detrend and induce stationary so that the real complexity of data can be captured rather than its seasonal cycles or trends. In this study, the data is detrended by taking the difference of a time data and its previous point. Other common ways adopted by researchers are using log-ratio transformation[a], and Gaussian kernel smoothing function[b,c]. Before applying it to different complexity measures, one can visualise the abrupt changes in climate record just from the plot of differenced data (see Figure 3), where it shows that the differences of data are minimum (points settle around zero) during the last part of the data set which corresponds to the holocene epoch.

In[ ]:=
```
tsReg = TimeSeriesResample[tsRaw, 0.025,
    ResamplingMethod → {"Interpolation", InterpolationOrder → 0}]
RegularlySampledQ[tsReg]
```

Out[ ]=
```
TimeSeries[ Time: 0.0951 to 50.  Data points: 1996 ]
```

Out[ ]=
```
True
```

In[ ]:=
```
AgeReg = Reverse[tsReg["Times"]];
TemperatureReg = Reverse[tsReg["Values"]];
Length[AgeReg];
Length[TemperatureReg];
TableView[
    Table[{AgeReg[[i]], TemperatureReg[[i]]}, {i, Length[TemperatureReg]}],
    AllowedDimensions → {2, Length[TemperatureReg]}];
```

In[ ]:=
```
regulardataAll = Table[TemperatureReg[[i]], {i, Length[TemperatureReg]}];
Length[regulardataAll ]
L1 = ListPlot[regulardataAll , GridLines → Automatic, Joined → True,
   PlotRange → All, AxesLabel → {RawBoxes["datapoints"], RawBoxes["Temperature,C"]},
   PlotLabel → "Figure 2 GISP2 ice core (regular-spaced datapoints)",
   LabelStyle → {GrayLevel[0], Bold}, ImageSize → 800];
Show[{
  L1,
  Plot[-60, {x, 1412, 1484}, PlotStyle → Red, Filling → Axis],
  Plot[-60, {x, 1484, 1531}, PlotStyle → Blue, Filling → Axis]
 }]
```

Out[ ]= 1996

Out[ ]=



```
In[•]:= diffdataAll = Differences[regulardataAll];
        Length[diffdataAll]
        ListLinePlot[diffdataAll, Joined → False, Filling → Axis, PlotRange → All,
         AxesLabel → {RawBoxes["datapoints"], RawBoxes["Temperature,C"]},
         PlotLabel → "Figure 3 GISP2 ice core differenced data",
         LabelStyle → {GrayLevel[0], Bold}, ImageSize → 800]
Out[•]= 1995
```

Out[•]=



4. Time Series Analysis

In this section, recurrence plots are first plotted for the whole climate record to visualise the nature of the system, and a few measures of complexity based on the plot is carried out. Then, changes of state and complexity of the record are investigated using Hurst exponent, sample entropy and burstiness via a rolling window.

4.1 Recurrence plot and RQA

Recurrence of a system to its former states is the fundamental of many dynamical systems. [d] Recurrence plots (RPs) are graphical tools that depict the different occasions when dynamical systems visit the same region of phase space and hence revealing complex deterministic patterns in dynamical systems: [e]

$R_{i,j}$ (ϵ) = Θ (ϵ - || $\vec{x}_i - \vec{x}_j$||), i,j = 1,2,3,...,N

where N is the number of measured points $\vec{x}_i$, ϵ is a threshold distance, Θ (.) is a Heaviside function (i.e. Θ (x) = 0, if x <0, and Θ (x)=1 otherwise) and ||.|| is a norm. [d] The length of diagonal lines which is parallel to the main diagonal line, Line of Identity reveals the nature of the system. One should expect long diagonal lines for predictable system, and short diagonal lines or even isolated points for stochastic system. [e] Selection of an appropriate ϵ has been a key question in Recurrence Quantification Analysis (RQA), and the fact that it strongly depends on the considered system [d], a value of ϵ = 0.15 standard deviation of climate time series is adopted in this study, based on the suggestion in [f]. The Recurrence Rate (RR) and Determinism (DET) are then calculated by using the Mathematica code adapted from [g]. The former estimates the probability that a certain state recurs while the latter provides an indication of determinism and predictability in the system. [e]

In[ ]:=
```
0.15 * StandardDeviation[regulardataAll]
```

Out[ ]=
```
1.11145
```

In[ ]:=
```
ϵ = 1.1;

g1 = MatrixPlot[Table[1 - UnitStep[ϵ - Norm[regulardataAll⟦t⟧ - regulardataAll⟦τ⟧]],
     {t, 1, Length[regulardataAll]}, {τ, 1, Length[regulardataAll]}],
   Mesh → False, FrameLabel → TraditionalForm /@ {t, τ}, PlotLabel → "Full"];

g2 = MatrixPlot[Table[1 - UnitStep[ϵ - Norm[regulardataAll⟦t⟧ - regulardataAll⟦τ⟧]],
     {t, 1, 600}, {τ, 1, 600}], Mesh → False,
   FrameLabel → TraditionalForm /@ {t, τ}, PlotLabel → "Top Left (1-600 datapoints)"];

g3 = MatrixPlot[Table[1 - UnitStep[ϵ - Norm[regulardataAll⟦t⟧ - regulardataAll⟦τ⟧]],
     {t, 600, 1320}, {τ, 600, 1320}], Mesh → False,
   FrameLabel → TraditionalForm /@ {t, τ}, PlotLabel → "Centre (600-1320 datapoints)"];

g4 = MatrixPlot[Table[1 - UnitStep[ϵ - Norm[regulardataAll⟦t⟧ - regulardataAll⟦τ⟧]],
     {t, 1200, Length[regulardataAll]},
     {τ, 1200, Length[regulardataAll]}], Mesh → False, FrameLabel →
     TraditionalForm /@ {t, τ}, PlotLabel → "Bottom Right (1200-1996 datapoints)"];

Show[ GraphicsGrid[{{g1, g2}, {g3, g4}}],
 PlotLabel → Style["Figure 4 Recurrence Plot with distance = 1.1", "Subsubsection"]]
```

Figure 4 Recurrence Plot with distance = 1.1

Out[ ]=



In[ ]:=
```
g5 = MatrixPlot[Table[Norm[regulardataAll⟦t⟧ - regulardataAll⟦τ⟧],
    {t, 1, Length[regulardataAll]}, {τ, 1, Length[regulardataAll]}], Mesh → False,
   FrameLabel → TraditionalForm /@ {t, τ}, PlotLabel → "Original Data"];

g6 = MatrixPlot[Table[Norm[diffdataAll⟦t⟧ - diffdataAll⟦τ⟧],
    {t, 1, Length[diffdataAll]}, {τ, 1, Length[diffdataAll]}], Mesh → False,
   FrameLabel → TraditionalForm /@ {t, τ}, PlotLabel → "Differenced Data"];

Show[GraphicsGrid[{{g5, g6}}],
 PlotLabel → Style["Figure 5 Recurrence Plot with all distances", "Subsubsection"]]
```

Out[ ]=

Figure 5 Recurrence Plot with all distances



From Figure 4, it is observed that the temperature is characterised by larger distances in the upper left quadrant (the big square which covers most of the area) of the plot, and small distances in the lower right corner. This suggests that these indices underwent a notable evolution in the period covered by the data. The lower right quadrant in fact covers the Holocene period data points, where the temperature became warmer. According to [e], “butterfly” shaped structure on RP is unique and its position might be related to some sharp changes. From Figure 4 bottom right plot, there is a “butterfly” recurrence structure occurs at the middle upper left, and the position of this shape in fact is the Bølling Allerød warming period.

Looking at the RP with all distances (see Figure 5), two distinct squares of the plots again revealed that the transition of temperature between the squares, and the bands around point 1500 from original data plot could be related to both Bølling Allerød warming and the end of Younger Dryas (starting of Holocene).

```
In[•]:= ϵ = 1.1;
rp = Table[UnitStep[ϵ - Norm[regulardataAll⟦t⟧ - regulardataAll⟦τ⟧]],
   {t, 1, Length[regulardataAll]}, {τ, 1, Length[regulardataAll]}];
n = Length[regulardataAll];
countlines = 0;
countpoints = 0;
For[j = 0, j < n, j++,
  linelength = 0;
  For[i = 1, i + j < n, i++,
   If[rp[[i, i + j + 1]] == 1, linelength++,
     If[linelength ≥ 2, countlines++,
      If[linelength == 1, countpoints++
       ];
     ];
     linelength = 0(*reset for a diagonal*)
    ];
  ];
  If[linelength ≥ 2, countlines++,
   If[linelength == 1, countpoints++
    ];
  ];
 ];
pr = N[(Total[Total[rp]]) / (n * n)];
pd = N[2 * countlines / (n * n)];
dr = N[pd / pr];
Print[StringJoin["Recurrence Rate  (RR): ", ToString[pr]]];
Print[StringJoin["Determinisim(DET): ", ToString[dr]]];

Recurrence Rate  (RR): 0.121143

Determinisim(DET): 0.164696
```

As reported from the code, the RR and DET of the system are 0.12 and 0.16 respectively. These measures are related to the length of diagonal lines in the recurrence matrix [d], hence, low values actually indicate that the low probability of a certain state recurs in this system and so the low determinism and predictability of the system, which is expected to see for climate records.

4.2 Hurst Exponent

Hurst Exponent is a measure of long-term memory of time series. If 1>H>0.5,there are long range time correlations, for 0.5>H>0, the series has long range anti-correlations, and if H=1.0, the process is deterministic while uncorrelated noise or random walk corresponds to H=0.5 [h]. The direct relationship of Hurst exponent and fractal dimension (which gives a measure of the roughness of a surface) is known by D = 2- H. The predictability index of temperature is then defined by $PI_T = 2 \mid D - 1.5 \mid$. [h] Interpretation of this index is simple: a value close to 0 indicates the corresponding process approximates the usual Brownian motion (as D = 1.5) and is therefore unpredictable while close to 1, the process is said to be very predictable. In this study, the single value of Hurst Exponent for whole climate record is first calculated. Then, a moving window of size 50 (which is approximately the length of Younger Dryas event) is applied for the record to detect the change of persistency in the data.

```
In[ ]:= windowPlot[data_, window_, f_, plegend_, color_, plabel_, xlabel_, ylabel_] :=
 Module[{movingmap, xAxis, windowedData, movingaverage, graph1, graph2, plot},
  movingmap = MovingMap[f, data, window];
  xAxis = Range[window, window + Length[movingmap]];
  windowedData = Table[{xAxis[[i]], movingmap [[i]]}, {i, Length[movingmap ]}];
  movingaverage = MovingAverage[windowedData, window];
  graph1 = ListLinePlot[windowedData, GridLines → Automatic, PlotStyle → {color},
    PlotRange → {{0, Length[data]}, Full}, PlotLegends → {plegend}];
  graph2 = ListLinePlot[movingaverage, PlotStyle → {Green},
    PlotLegends → {"Averange value"}];
  plot = Show[graph1, graph2, AxesLabel → {RawBoxes[xlabel], RawBoxes[ylabel]},
    PlotLabel → plabel, LabelStyle → {GrayLevel[0], Bold}, ImageSize → 800];
  plot]

In[ ]:= hurstExp[data_] := h /. (FindProcessParameters[data, FractionalBrownianMotionProcess[h]])

In[ ]:= Hvalue = hurstExp[regulardataAll]
Dfractal = 2 - Hvalue
Pindex = 2 * Abs[Dfractal - 1.5]

Out[ ]= 0.841255

Out[ ]= 1.15875

Out[ ]= 0.682509
```

The Hurst for the whole record is 0.84, which tells the long-term memory and high persistence of the temperatures in Greenland before present. The Fractal Dimension is 1.16 and the Predictability Index is 0.68, suggests that the temperature data is far from random but also unpredictable. Figure 6 shows the moving window analysis of Hurst exponent over the data. From the figure, it is observed that in overall the time series exhibits reinforcing behaviour through time, with great fluctuations of persistency during the DO event and becomes relatively stable during the LGM. A final jump of H-value is seen onset of Holocene period, and the temperature depicts a very strong correlated behaviour entering the Holocene epoch.

In[ ]:=
```
windowPlot[regulardataAll, 50, hurstExp, "Original data",
  Blue, "Figure 6 Hurst Exp over time", "datapoints", "Hurst Exponent"]
```

Out[ ]=



4.3 Sample Entropy

Sample entropy (SampEn) is one of the regularity statistics been widely used to quantify randomness and complexity and is useful to classify systems and applicable to stochastic and deterministic processes [a]. Larger values of denote higher complexity and lower values imply organization and a higher degree of predictability. The Mathematica code applied is adapted from [i]. Similar to the previous part, a value of SampEn for whole data is first calculated, and followed by the rolling window time series. In order to avoid the cyclic behavior of the temperature influencing the obtained results, and the usefulness of SampEn to determine the complexity could be compromised, it is suggested by [j] to use the stationary version of data rather than the raw data. Two parameters, embedding dimension or the size of the template being compared, m and noise filter r where points within a distance of r are considered equal, are set to 2 and 0.2σ respectively, as suggested in [a,j].

In[◦]:=
```
SampEn[data_] := Module[{m, r, nF1, nF2, diff1, diff2, va1, va2},
  m = 2;
  r = 0.2 * StandardDeviation[data];
  va1 = Partition[data, m, 1];
  va2 = Partition[data, 1 + m, 1];
  nF1 = Nearest[va1 → Automatic, DistanceFunction → ChessboardDistance];
  nF2 = Nearest[va2 → Automatic, DistanceFunction → ChessboardDistance];
  diff1 = Total[(Length /@ nF1[va1, {All, r}])] - Length[va1];
  diff2 = Total[(Length /@ nF2[va2, {All, r}])] - Length[va2];
  -Log[N@(diff2 / diff1)]]
```

In[◦]:=
```
SampEn[diffdataAll]
```

Out[◦]= 1.17748

In[◦]:=
```
windowPlot[diffdataAll, 50, SampEn, "Differenced data", Red,
  "Figure 7 Sample Entropy over time", "datapoints", "Sample Entropy"]
```

Out[◦]=



Using a window of 1250 ages, the trend in Figure 7 shows that the complexity of the temperature in Greenland from 50 thousand ages before present fluctuates around 1.25 and increased slightly when entering LGM before a drop. The complexity then increases by time and eventually reaches a peak of 2.8 at the end of Bølling Allerød warming. This peak is then followed by a plummet to the lowest value at 0.35. Tracing back to the time series, this is actually the time after entering Holocene epoch, where the temperature raises linearly before stabilised at around -30°C.

4.4 Burstiness

Burstiness is a measure of the dispersion of a probability distribution where high values represent high volatility relative to the mean and therefore high complexity. [k] One of the measures of burstiness is the Fano Factor like the coefficient of variation, which is a ratio between the variance and the mean or absolute mean. [l,m] In this study, burstiness is measured using a modified coefficient of variation, $\bar{c_v}$ from [k]:

$\bar{c_v} = (c_v - 1)/(c_v + 1)$, where $c_v = \sigma / |\mu|$

In[ ]:=
```
burstiness[data_] := Module[{std, mean, coeff, burst},
  std := StandardDeviation[data];
  mean := Abs[Mean[data]];
  coeff := std / mean;
  burst := (coeff - 1) / (coeff + 1);
  burst]
```

In[ ]:=
```
burstiness[diffdataAll]
```

Out[ ]=
```
0.987912
```

In[ ]:=
```
windowPlot[diffdataAll, 50, burstiness, "Differenced data",
  Red, "Figure 8 Burstiness over time", "datapoints", "Burstiness"]
```

Out[ ]=


The burstiness of the data is 0.987. Figure 8 reveals the changes in burstiness of stationary data rolling over a window size of 50. The trend of complexity based on burstiness changes and fluctuates sharply before LGM, and becomes relatively stable and high after LGM. It is noticed that the burstiness begins to drop while approaching the Bølling Allerød warming, and continue to decrease until after the end of Younger Dryas. The complexity shoots up after reaching the trough, and relatively stable at value above 0.8 during the Holocene period.

5. Critical Slowing Down

In this section, the climate record is cropped so that the dynamics and transitions of temperatures during Bolling Allerod Warming and end of Younger Dryas can be better captured. The aim is to use several indicators to detect early warning signals of those well-known abrupt climate changes during last deglaciation. The idea is based on the generic phenomenon called "critical slowing down", that in the vicinity of many kinds of tipping points, the recovery rate of a system from small perturbations becomes very slow. [n] It can be inferred indirectly from rising "memory" in small fluctuations in the state of a system, as reflected, for instance, in a higher lag-1 autocorrelation, increased variance, change in skewness and kurtosis. In this study, the moving window size is taken as the half of the number of data points, as suggested in [b].

In[ ]:=
```
tsPart = TimeSeriesWindow[tsReg, {10, 21}]
Show[
 ListLinePlot[tsPart,
  ScalingFunctions → {"Reverse", Identity}, PlotRange → {{10, 21}, All},
  AxesLabel → {RawBoxes["time, x10^3 yrs BP"], RawBoxes["Temperature,∘C"]},
  PlotLabel → "Figure 10 LGM to end of Younger Dryas (time series)",
  LabelStyle → {GrayLevel[0], Bold}, ImageSize → 800],
 Plot[-60, {x, -12.9, -11.7}, PlotStyle → Blue, Filling → Axis],
 Plot[-60, {x, -14.69, -12.9}, PlotStyle → Red, Filling → Axis]
]
```

Out[ ]=
```
TimeSeries[ Time: 10. to 21.
            Data points: 440 ]
```

Out[ ]=


Figure 10 LGM to end of Younger Dryas (time series)
20
18
16
14
12

In[ ]:=
```
AgePart = Reverse[tsPart["Times"]];
TemperaturePart = Reverse[tsPart["Values"]];
Length[AgePart]
Length[TemperaturePart]
regulardataPart = Table[TemperaturePart[[i]], {i, Length[TemperaturePart]}];
TableView[Table[{AgePart[[i]], TemperaturePart[[i]]}, {i, Length[TemperaturePart]}],
 AllowedDimensions → {2, Length[TemperaturePart]}]
```

Out[ ]= 440

Out[ ]= 440

Out[ ]=

	1	2
1	20.9951	−48.4609
2	20.9701	−48.0428
3	20.9451	−48.0428
4	20.9201	−47.6248
5	20.8951	−47.6248
6	20.8701	−47.2272
7	20.8451	−47.2272
8	20.8201	−47.4719
9	20.7951	−47.4719
10	20.7701	−46.585
11	20.7451	−46.585
12	20.7201	−45.5862
13	20.6951	−45.5862
14	20.6701	−44.5055
15	20.6451	−44.5055

In[ ]:=
```
Show[
 ListPlot[regulardataPart, GridLines → Automatic, Joined → True, PlotRange → All,
  AxesLabel → {RawBoxes["datapoints"], RawBoxes["Temperature,∘C"]},
  PlotLabel → "Figure 12 LGM to end of Younger Dryas (simple data)",
  LabelStyle → {GrayLevel[0], Bold}, ImageSize → 800],
 Plot[-60, {x, 325, 372}, PlotStyle → Blue, Filling → Axis],
 Plot[-60, {x, 252, 325}, PlotStyle → Red, Filling → Axis]
]
```

Out[ ]=



In[ ]:=
```
Histogram[regulardataPart, 100, PlotRange → All,
  AxesLabel → {RawBoxes["Temperature,∘C"], RawBoxes["Frequency"]}, GridLines → Automatic,
  PlotLabel → "Figure 13 Histogram", LabelStyle → {GrayLevel[0], Bold}, ImageSize → 800]
```

Out[ ]=



In[ ]:=
```
diffdataPart = Differences[regulardataPart];
Length[diffdataPart]
ListLinePlot[diffdataPart, Joined → False, Filling → Axis, PlotRange → All,
 AxesLabel → {RawBoxes["datapoints"], RawBoxes["Temperature,∘C"]},
 PlotLabel → "Figure 14 LGM to end of Younger Dryas (differenced data)",
 LabelStyle → {GrayLevel[0], Bold}, ImageSize → 800]
```

Out[ ]= 439

Out[ ]=



5.1 Variance

**Note that rolling window plot is modified such that it shows the events where abrupt changes occur, with red dashed line indicates the start of Bølling Allerød warming, and blue dashed line denotes the end of Younger Dryas. Last 25 data points are also been cropped so that the resultant graph focus more on the changes before transitions.

In[]:=
```
windowPlot2[data_, window_, f_, plegend_, color_, plabel_, xlabel_, ylabel_] :=
 Module[{movingmap, xAxis, windowedData, graph, plot},
  movingmap = MovingMap[f, data[[1 ;; Length[data] - 25]], window];
  xAxis = Range[window, Length[data]];
  windowedData = Table[{xAxis[[i]], movingmap [[i]]}, {i, Length[movingmap ]}];
  graph = ListLinePlot[windowedData, GridLines → {{{372, Directive[Blue, Thick, Dashed]},
       {252, Directive[Red, Thick, Dashed]}}, Automatic}, PlotStyle → {color},
    PlotRange → {{0, Length[data]}, Full}, PlotLegends → {plegend}];
  plot = Show[graph, AxesLabel → {RawBoxes[xlabel], RawBoxes[ylabel]},
    PlotLabel → plabel, LabelStyle → {GrayLevel[0], Bold}, ImageSize → 800];
  plot]
```

In[]:=
```
windowPlot2[diffdataPart, 220, StandardDeviation, "Differenced data", Red,
 "Figure 15 Standard deviation over time", "datapoints", "Standard deviation"]
```

Out[]=



Discussion of the result in this part will be discussed together with next part (5.2).

5.2 Autocorrelation

In[]:=
```
ACF1[data_] := CorrelationFunction[data, 1];
```

In[ ]:=
```
windowPlot2[diffdataPart, 220, ACF1, "Differenced data", Red,
 "Figure 16 Autocorrelation Order 1 over time", "datapoints", "AC1 coeff"]
```

Out[ ]=



Both variance and autocorrelation graphs show a clear increase trend towards the end of of Younger Dryas (see Figure 15 and 16). As the window size is very large (half of the length of data), it is hard to tell, at least from these figures, if there are any signs of slowing down for Bølling Allerød warming.

5.3 Skewness

The skewness is a dimensionless measure of the degree of asymmetry of a probability distribution in which it vanishes for distribution symmetric about the mean and is positive or negative for an asymmetric distribution with a tail above or below the mean respectively. [o] The authors in [o] suggested that it is the change in skewness which acts as an early warning signal, and could be from zero skewness to either positive or negative values or from one sign of skewness to the other.

In[ ]:=
```
windowPlot2[diffdataPart, 220, Skewness, "Differenced data",
 Red, "Figure 17 Skewness over time", "datapoints", "Skewness"]
```

Out[ ]=



There is an increase in and hence a change of skewness from negative to positive values for Bølling Allerød warming (see Figure 17). The skewness of the system shows a decrease in general before the end of Younger Dryas, however, the values are still positive. Hence, based on the interpretation from [o], using skewness as an indicator shows an early warning signal for Bølling Allerød warming but not end of Younger Dryas.

5.4 Kurtosis

As mentioned, approaching tipping points, the variance of the system increases. Thus it enhances the tail of the distribution. Kurtosis is the measure of the tailedness of the distribution, so it is also used as an early warning signal for predicting tipping points. [p] It is expected to see a decrease in kurtosis near tipping points. It is observed that the kurtosis of system in fact shows a drop near the period of the end of Younger Dryas (see Figure 18).

In[ ]:=
```
windowPlot2[diffdataPart, 220, Kurtosis, "Differenced data",
  Red, "Figure 18 Kurtosis over time", "datapoints", "Kurtosis"]
```

Out[ ]=



6. Comparative Analysis of dataset

In this section, the real time series is detrended by using Gaussian kernel with a range of smoothing levels, r. The purpose is to apply the complexity measures in Section 4 such as SampEn and burstiness to the smoothed data, as a comparative analysis of complex time series with linear data. The comparative analyses of data are carried out by plotting measures of complexity value vs degree of smoothing. In the end of this section, a RP is constructed for the data with highest level of r, again, as a comparison with the original time series. Note that in Section 2, stationary version of data is used in application of SampEn and burstiness, hence, differenced data of linear data is also employed for the mentioned measures.

6.1 Detrend time series using Gaussian Kernel Filter

This part is to show the higher the smoothing factor, the more linear data is produced.

In[ ]:=
```
res = Table[GaussianFilter[tsReg, r], {r, {10, 25, 50, 200, 1000}}];
ListLinePlot[Join[{tsReg}, res], ScalingFunctions → {"Reverse", Identity},
 PlotLegends → {"data", "r = 10", "r = 25", "r = 50", "r = 200", "r = 1000"},
 AxesLabel → {RawBoxes["time, x10³ yrs BP"], RawBoxes["Temperature,∘C"]},
 PlotLabel → "Figure 19 Gaussian Kernel Filter detrending",
 LabelStyle → {GrayLevel[0], Bold}, ImageSize → 800]
```

Out[ ]=



6.2 Measures of complexity of linear data

6.2.1 Sample Entropy and burstiness

In[ ]:=
```
SampEncompare =
  Table[SampEn[Differences[GaussianFilter[tsReg, r]["Values"]]], {r, 1, 1001, 20}];
rValues = Range[1, 1001, 20];
dataSampEn = Transpose@{rValues, SampEncompare};
s1 = ListLinePlot[dataSampEn , PlotRange → All,
   GridLines → Automatic, PlotStyle → {Red}, PlotLegends → {"sample entropy"}];

burstcompare =
  Table[burstiness[Differences[GaussianFilter[tsReg, r]["Values"]]], {r, 1, 1001, 20}];
databurst = Transpose@{rValues, burstcompare};
b1 = ListLinePlot[databurst, PlotRange → All,
   GridLines → Automatic, PlotStyle → {Blue}, PlotLegends → {"burstiness"}];

Show[{s1, b1}, PlotRange → All,
 AxesLabel → {RawBoxes["smoothing factor r"], RawBoxes["Complexity measures"]},
 PlotLabel → "Figure 20 Measures of complexity vs smoothing factor",
 LabelStyle → {GrayLevel[0], Bold}, ImageSize → 800]
```

Out[ ]=



Figure 20 depicts that the higher linearity, the lower complexity of data in terms of sample entropy and burstiness. It also justifies that the original data (without any degree of detrending) is indeed complex and non-linear.

6.2.2 Recurrence plot

In[ ]:=

```
(*Extract one timeseries from Section 6.1*)
lineardata = res[[5]]["Values"]; (*when r = 1000*)
0.15 * StandardDeviation[lineardata]
```

Out[ ]=

```
0.573421
```

```
In[ ]:= ϵ = 0.01;

g7 = MatrixPlot[Table[1 - UnitStep[ϵ - Norm[leardata[[t]] - leardata[[τ]]]],
     {t, 1, Length[leardata]}, {τ, 1, Length[leardata]}],
   Mesh → False, FrameLabel → TraditionalForm /@ {t, τ}, PlotLabel → "Full"];

g8 = MatrixPlot[Table[1 - UnitStep[ϵ - Norm[leardata[[t]] - leardata[[τ]]]],
     {t, 1, 600}, {τ, 1, 600}], Mesh → False,
   FrameLabel → TraditionalForm /@ {t, τ}, PlotLabel → "Top Left (1-600 datapoints)"];

g9 = MatrixPlot[Table[1 - UnitStep[ϵ - Norm[leardata[[t]] - leardata[[τ]]]],
     {t, 600, 1320}, {τ, 600, 1320}], Mesh → False,
   FrameLabel → TraditionalForm /@ {t, τ}, PlotLabel → "Centre (600-1320 datapoints)"];

g10 = MatrixPlot[Table[
     1 - UnitStep[ϵ - Norm[leardata[[t]] - leardata[[τ]]]], {t, 1200, Length[leardata]},
     {τ, 1200, Length[leardata]}], Mesh → False, FrameLabel → TraditionalForm /@ {t, τ},
   PlotLabel → "Bottom Right (1200-1996 datapoints)"];

Show[GraphicsGrid[{{g7, g8}, {g9, g10}}], PlotLabel →
  Style["Figure 21 Recurrence Plot with distance = 0.01 (linear data)", "Subsubsection"]]
```

Out[ ]=

Figure 21 Recurrence Plot with distance = 0.01 (linear data)



```
In[ ]:= ϵ = 0.01;
rp = Table[UnitStep[ϵ - Norm[lineardata[[t]] - lineardata[[τ]]]],
   {t, 1, Length[lineardata]}, {τ, 1, Length[lineardata]}];
n = Length[lineardata];
countlines = 0;
countpoints = 0;
For[j = 0, j < n, j++,
  linelength = 0;
  For[i = 1, i + j < n, i++,
   If[rp[[i, i + j + 1]] == 1, linelength++,
     If[linelength ≥ 2, countlines++,
      If[linelength == 1, countpoints++
       ];
     ];
     linelength = 0(*reset for a diagonal*)
    ];
  ];
  If[linelength ≥ 2, countlines++,
   If[linelength == 1, countpoints++
    ];
  ];
 ];
pr = N[(Total[Total[rp]]) / (n * n)];
pd = N[2 * countlines / (n * n)];
dr = N[pd / pr];
Print[StringJoin["Recurrence Rate (RR): ", ToString[pr]]];
Print[StringJoin["Determinisim(DET): ", ToString[dr]]];

Recurrence Rate (RR): 0.0044794

Determinisim(DET): 0.088087
```

Figure 21 shows the recurrence plot of detrended time series with a factor of 1000, a big contrast with what is observed in Figure 4, where no patterns are seen. The RR and DET measured are 0.004 and 0.09 respectively, which are much lower than the values before (0.12 and 0.16).

7. Conclusion

In conclusion, in time series analysis, the dynamics and complexity of whole paleoclimate record from Greenland ice core are studied, and almost all of them show significant transition either in pattern (RP) or values (Hurst, SampEn, Burstiness) at the end of Younger Dryas. For detection of critical slowing down in period from Last Glacial Maximum to the end of Younger Dryas, strong signs of early warning are spotted for the end of Younger Dryas using indicator such as standard deviation, autocorrelation and kurtosis, while the abrupt change in warming of Bølling Allerød has only one positive result with indicator skewness. These results are somewhat parallel to previous studies, where [b,c] had found a positive trend in variability and autocorrelation for the end of Younger Dryas and a weak early sign in autocorrelation for the starting of Bølling Allerød. Then, the comparative analysis of the time series affirms its non-linearity nature. Finally, it is suggested to extend the complexity analysis by diving deeper into RQA. Comparison of Approximate Entropy and Sample Entropy on current data can also be carried out as in [a,j].

References

[1] https://en.wikipedia.org/wiki/Deglaciation

[2] https://en.wikipedia.org/wiki/Dansgaard%E2%80%93Oeschger_event

[3] https://en.wikipedia.org/wiki/B%C3%B8lling%E2%80%93Aller%C3%B8d_warming

[4] https://en.wikipedia.org/wiki/Younger_Dryas

[5] https://en.wikipedia.org/wiki/Last_Glacial_Maximum

[6] Lenton, T. & Livina, Valerie & Dakos, Vasilis & Scheffer, M.. (2012). Climate bifurcation during the last deglaciation. Climate of the Past Discussions. 8. 321-348. 10.5194/cpd-8-321-2012.

[7] Alley R (2004) GISP2 Ice Core Temperature and Accumulation Data. IGBP PAGES/World .Data Center for Paleoclimatology Data Contribution Series no. 2004-013.
NOAA/NGDC Paleoclimatology Program, Boulder CO.

[a] Delgado-Bonal, A., Marshak, A., Yang, Y. et al. Analyzing changes in the complexity of climate in the last four decades using MERRA-2 radiation data. Sci Rep 10, 922 (2020). https://doi.org/10.1038/s41598-020-57917-8

[b] Dakos, Vasilis & Scheffer, Marten & Nes, Egbert & Brovkin, Victor & Petoukhov, V. & Held, Hermann. (2008). Slowing down as an early warning signal for abrupt climate change. Proceedings of the National Academy of Sciences of the United States of America. 105. 14308-12. 10.1073/pnas.0802430105.

[c] Lenton, T & Livina, Valerie & Dakos, Vasilis & Nes, E & Scheffer, M. (2012). Early warning of climate tipping points from critical slowing down: Comparing methods to improve robustness. Philosophical transactions. Series A, Mathematical, physical, and engineering sciences. 370. 1185-204. 10.1098/rsta.2011.0304.

[d] Marwan, Norbert & Romano, Maria & Thiel, Marco & Kurths, Juergen. (2007). Recurrence Plots for the Analysis of Complex Systems. Physics Reports. 438. 237-329. 10.1016/j.physrep.2006.11.001.

[e] Bastos, João & Caiado, Jorge. (2011). Recurrence quantification analysis of global stock. Physica A-statistical Mechanics and Its Applications - PHYSICA A. 390. 10.1016/j.physa.2010.12.008.
[f] Zhang, Wen & Feng, Guoling & Liu, Qunqun. (2016). The Application of Recurrence Quantification Analysis in Detection of Abrupt Climate Change. Discrete Dynamics in Nature and Society. 2016. 1-7. 10.1155/2016/2689429.
[g] Güldal, Serkan & Baugh, Veronica. (2015). Dynamic System Analysis Using Recurrence Quantification Analysis and Recurrence Plots In Mathematica.
[h] Rehman S, Study of Saudi Arabian climatic conditions using Hurst exponent and climatic predictability index, Chaos, Solitons & Fractals (2007), doi:10.1016/j.chaos.2007.01.079
[i] https://mathematica.stackexchange.com/questions/162634/can-this-sample-entropy-sampen-calculation-be-improved
[j] Delgado-Bonal A, Marshak A. Approximate Entropy and Sample Entropy: A Comprehensive Tutorial. Entropy. 2019; 21(6):541.
[k] Michael Harré (2019). Highly rational to wildly irrational: Housing markets are a heterogeneous ensemble.
[l] https://en.wikipedia.org/wiki/Burstiness
[m] https://en.wikipedia.org/wiki/Coefficient_of_variation
[n] Scheffer, Marten & Carpenter, Stephen & Lenton, Timothy & Bascompte, Jordi & Brock, William & Dakos, Vasilis & van de Koppel, Johan & van de Leemput, Ingrid & Levin, Simon & Nes, Egbert & Pascual, Mercedes & Vandermeer, John. (2012). Anticipating Critical Transitions. Science (New York, N.Y.). 338. 344-8. 10.1126/-science.1225244.
[o] Guttal, V. & Jayaprakash, Ciriyam. (2008). Changing skewness: an early warning signal of regime shift in ecosystems. Ecology Letters. 201. 420-428.
[p] Nazarimehr, Fahimeh & Jafari, Sajad & Golpayegani, Seyed & Perc, Matjaž & Sprott, Julien Clinton. (2018). Predicting tipping points of dynamical systems during a period-doubling route to chaos. Chaos: An Interdisciplinary Journal of Nonlinear Science. 28. 073102. 10.1063/1.5038801.